

Measuring the membrane porosity on microalgae (Under progress)

Cinquin¹ and sakina.bm²

¹NEUBIAS

²Affiliation not available

May 26, 2019

Cinquin Bertrand, Bensalem Sakina

Introduction :

Seen as microfactories, microalgae are organisms already in use in many industries. However, the molecules of interests they're making are enclosed and so far the only way of releasing is done by killing the biomass, filtering and extracting the molecule. Working on membrane porosity is challenging and could help finding a "clean way" of producing such molecules. Controlled electroporation could be a way. Previous experiments shows that there is a voltage dependency to form non permanent holes in the membrane. This reversible damage will allow the escape of some molecules and the repair by the microalgae to take place. However, measuring the effects on electroporation different parameters is challenging.

Keyword : Imagej Macro Langage

The experiment :

Microalgae are electropored and bathed into a fluorescent dextran solution of different molecular weight. If holes are created and are big enough, dextran molecules will enter the microalgae and fluorescence will be measured. Acquisition is performed on a confocal microscope. Single plane is acquired (at the center of the microalgae).

We acquire different channel : Dextran (green fluorescence), Membrane (tagged in yellow fluorescence XXX), Chlorophyll (Autofluorescence in red), visible channel (for visualization of microalgae integrity).

The workflow :

We will use the Membrane channel to segment the microalgae, report the mask on the Dextran channel and realise concentric measurements.




Figure 1: Example of the four acquired channels (scale bar is 10 μm)




Figure 2: Workflow view

1. Step 1 Ordering and opening the images

We use the plugin Bio-Formats Macro Extensions to be able to open only the images we want to treat, saving time and space. To know exactly which ones, primary knowledge of the file architecture is necessary .

As we want to work on all the series of the data, we need to retrieve the number of series. We name the variable seriesCount. Then we will open only the images using a “for loop” selecting the series using Ext.setSeries(i) and opening the right channel with Ext.openImage(“title”, x). Where x is the index of the desired channel. 0, the first channel for the dextran and 2, the third channel for the membrane.

Then we run Images to Stack for each channel ordering them in two separate multitiff images stack, one called DextranChannel and the second one MembraneChannel.

Image to Stack operation lost the scaling factor. We restate it with the function “Set Scale...”

```

\begin{lstlisting}
Dir1 = getDirectory("Choose Source Directory");
ImageList =getFileList(Dir1);
Array.show(ImageList);
FileName =Dir1+ImageList[0];print(FileName);

```

```

run("Bio-Formats Macro Extensions");
Ext.setId(fileName);
Ext.fileGroupOption(fileName, must);
Ext.getSeriesCount(seriesCount); print(seriesCount);
for (i=0;i<seriesCount;i++) {
    Ext.setSeries(i);
    Ext.openImage("title", 0);
}
run("Images to Stack", "method=[Scale (largest)] name=DextranChannel title=[] use");
run("Set Scale...", "distance=1 known=0.18054 pixel=1 unit=micron");
for (i=0;i<seriesCount;i++) {
    Ext.setSeries(i);
    Ext.openImage("title", 2);
}
run("Images to Stack", "method=[Scale (largest)] name=MembraneChannel title=[] use");
run("Set Scale...", "distance=1 known=0.18054 pixel=1 unit=micron");
\end{lstlisting}

```

`\textbf{sourcecode}` [\href{https://github.com/Arktthul/Measuring-the-membrane-porosity-on-microalgae-Under-progress/blob/master/MicroAlgues_Part01.ijm}](https://github.com/Arktthul/Measuring-the-membrane-porosity-on-microalgae-Under-progress/blob/master/MicroAlgues_Part01.ijm){< MicroAlgues_Part01.ijm>}

sourcecode : [part 1](#)

2. Step 2 : Segmentation

Segmentation occurs in two major steps. Preparation of the MembraneChannel and Segmentation itself.

1. As the membrane labelling can have a noisy background, we realise this series of action

Duplicate the MembraneChannel image

Run a gaussian blur

Realise the subtraction of MembraneChannel by this new image we called BackgroundCleanup.

Close the image to save space

Rename the result of subtraction to be called ForSegmentation.

```

\begin{lstlisting}
run("Duplicate...", "title=BackgroundCleanup duplicate");
run("Gaussian Blur...", "sigma=50 stack");
imageCalculator("Subtract create stack", "MembraneChannel", "BackgroundCleanup");
close("BackgroundCleanup");
selectWindow("Result of MembraneChannel");rename("ForSegmentation");
\end{lstlisting}

```

sourcecode https://github.com/Arktthul/Measuring-the-membrane-porosity-on-microalgae-Under-progress/blob/master/MicroAlgues_Part01.ijm {< MicroAlgues_Part02.ijm>}

sourcecode : [part 2](#)

2. The segmentation itself is realised like that

Two simple smoothing

Apply a global threshold between 8, and 255. We remove only little low intensity specks

We define the measurements we are going to extract

We run “Analyse Particles” with the following parameters that could be tuned differently

size between 20 and Infinity

circularity between 0.2 and 1.00.

Then, we rename the table named Results (by default), by TableResultofSegmentation.

```
\begin{lstlisting}
run("Smooth", "stack");run("Smooth", "stack");
setThreshold(8, 255);
run("Set Measurements...", "area mean bounding fit shape stack display redirect=None decimal=3");
run("Analyze Particles...", "size=20-Infinity circularity=0.2-1.00 display add stack");
selectWindow("Results");Table.rename("Results","TableResultofSegmentation");
\end{lstlisting}
```

sourcecode https://github.com/Arktthul/Measuring-the-membrane-porosity-on-microalgae-Under-progress/blob/master/MicroAlgues_Part01.ijm {< MicroAlgues_Part03.ijm>}

sourcecode : [part 3](#)

Remark : Circularity could be set up at a higher parameter but we most likely will loose some algae that are budding. It will depends if such criterion of sorting is necessary.

After this major step, we have a series of region of interest that we can use to extract the intensity information from the dextran channel.

3 Step 3 Extraction

This step seems certainly the easiest but will be in fact quite complex since we are working with stacks and several dozens of microalgae have been segmented.

We want to extract concentric intensity from each microalgae. As we can operate only one list of region of interest, we have to be careful with the indexes of the new ROIS that will be created. In addition to that, instead of having as many table as microalgae, we will create a finaltable in which each line will correspond to a single microalgae.

1. Initialization

```
\begin{lstlisting}
ROIcount =roiManager("count");
//init Final Table
run("Set Measurements...", " redirect=None decimal=3");
roiManager("Select", 0);run("Measure");Table.rename("Results", "FinalTable");
\end{lstlisting}
```

`\textbf{sourcecode}` [\href{https://github.com/Arktthul/Measuring-the-membrane-porosity-on-microalgae-Under-progress/blob/master/MicroAlgues_Part04.ijm}](https://github.com/Arktthul/Measuring-the-membrane-porosity-on-microalgae-Under-progress/blob/master/MicroAlgues_Part04.ijm){< MicroAlgues_Part04.ijm>}

sourcecode : [part 4](#)

2. Creation of concentric rings

Different methods can be applied to create concentric rings. Constant radius displacement, constant areas. This can be done with a normalisation by the size of the microalgae.

Each one can be interesting and supported by different diffusion models : free diffusion, viscosity model and depends of primay knowledge on the organism.

We approximate the microalgae as a circle for which the radius can be retrieved by measuring the area. We compute a variable Diff that will be depending on the circle number.

We decided to realise ten concentric rings for which each donut new Rois will have the same approximative area. This piece of code create ten new circles each of them slightly smaller.

```
\begin{lstlisting}
for (i=0; i<ROIcount; i++){
    selectWindow("DextranChannel");run("Set Measurements...", "area mean display redirect=None decimal=
    roiManager("Select", i);
    run("Measure");
    area1= getResult("Area",0);
    //print(area1);
    Radius = sqrt(area1/(3.14));
    print(Radius);
    roiManager("Add");
    for(j=1;j<10;j++){
        roiManager("Select",ROIcount);
        print(j, (10-j)/10,sqrt((10-j)/10)*Radius);
        Diff=Radius*(1-sqrt((10-j)/10));
        run("Enlarge...", "enlarge=-"+Diff);
        roiManager("Add");
    }
}
\end{lstlisting}
```

`\textbf{sourcecode}` [\href{https://github.com/Arktthul/Measuring-the-membrane-porosity-on-microalgae-Under-progress/blob/master/MicroAlgues_Part04.ijm}](https://github.com/Arktthul/Measuring-the-membrane-porosity-on-microalgae-Under-progress/blob/master/MicroAlgues_Part04.ijm){< MicroAlgues_Part05.ijm>}

sourcecode : [part 5](#)

To create the donnut shape ROI, we will select two consecutive circles and use XOR logical operator, the new shape will be added to the Roimanager.

Theoretically , it should work but the number of pixels is finite. The computation work in a continuous space but not in a discrete one. That means that Diff can be 0 or round up to 0 because the microalgae is too small to begin with. In this case, the new created circle will be the same as the previous one. By consequence, XOR of two same ROIS is void and no new ROI can be created. If we force the creation of a Roi, the macro will ceased to run as there is no active selection on the image. A way to avoid that is to measure after roiManager("XOR") and test the area. If Area equal the size of the window (it is of 184 μm by 184 μm that is 34178) that means that indeed XOR operation is void and the loop need to jump to the iteration.

That explains the first line of the following excerpt. We initialise a variable Testcount at 0. This variable will be decreased by 1 each time XOR is void and avoid the crash of the macro.

```
\begin{lstlisting}
Testcount = 0;
  for(k=1;k<10;k++){
    l = k+ROIcount-1;
    Selec =newArray(l,k+ROIcount);
    roiManager("Select",Selec);
    roiManager("XOR");
    run("Measure"); area2= getResult("Area",k+Testcount);
    if (area2 >= 34178){
      print("XOR selection is null");
      roiManager("Select",l);
      Table.deleteRows(k, k);
      Testcount = Testcount-1;
    }
    roiManager("Add");
  }
\end{lstlisting}
```

sourcecode [\href{https://github.com/Arktthul/Measuring-the-membrane-porosity-on-microalgae-Under-progress/blob/master/MicroAlgues_Part04.ijm}](https://github.com/Arktthul/Measuring-the-membrane-porosity-on-microalgae-Under-progress/blob/master/MicroAlgues_Part04.ijm){< MicroAlgues_Part06.ijm>}

sourcecode : [part 6](#)

3. Measure

For each microalgae, we will measure between 10+ROIcount until 19+Roicount. These are the indexes of the donnut new ROIS.

The last ROI is the central area of each microalgae, its index is 9+ROIcount.

We create a newArray called IntensityArray in which intensity of each Rois will be stored. This new result table is called ResultsofsingleMicroAlgae+i (index of the microalgae). Here we are closing it as the number of created tables will be too much to handle properly.

```
\begin{lstlisting}
  for(n=10+ROIcount;n<19+ROIcount;n++){
```

```

        roiManager("Select",n);
        run("Measure");
    }
    roiManager("Select",9+ROIcount);
    run("Measure");
    Label = getResultString("Label",0)+"_"+i+1; print(Label);
    IntensityArray = newArray(11);
    IntensityArray[0]=getResult("Mean",0);
    for (m=1;m<=10;m++){
        IntensityArray[m]=getResult("Mean",m+9+Testcount);
    }
    Table.rename("Results","ResultsofsingleMicroAlgae"+i);run("Close");
\end{lstlisting}

```

`\textbf{sourcecode}` `\href{https://github.com/Arktthul/Measuring-the-membrane-porosity-on-microalgae-Under-progress/blob/master/MicroAlgues_Part04.ijm}{< MicroAlgues_Part07.ijm>}`

sourcecode : [part 7](#)

4. Final Table

Some operations can be done on a result table only if the table is called “Results”. We will unprotect “Finaltable” in renaming the table “Results” to modify it and protect it again by switching the name back to “FinalTable”.

Final table has been initialised with as little information as we can during **step1 . initialization**

We will then creating proper labels for each column.

The first one is the name of the microalgae, the second one is its average intensity, the third to the 13th one will contain the intensity for each concentric rings. We named them “Intensity Slice”+n

Once the addition has been realised, we rename Results as FinalTable. The next microalgae will be able to be treated.

```

\begin{lstlisting}
    selectWindow("FinalTable");
    Table.rename("FinalTable","Results");
    setResult("Name", i, Label);
    ColumnTitle = "Average Intensity Algae";
    setResult(ColumnTitle, i, IntensityArray[0]);
    for (n=1;n<=10;n++){
        ColumnTitle = "Intensity Slice"+n;
        setResult(ColumnTitle, i, IntensityArray[n]);
    }
    Table.rename("Results","FinalTable");
\end{lstlisting}

```

`\textbf{sourcecode}` [\href{https://github.com/Arktthul/Measuring-the-membrane-porosity-on-microalgae-Under-progress/blob/master/MicroAlgues_Part04.ijm}](https://github.com/Arktthul/Measuring-the-membrane-porosity-on-microalgae-Under-progress/blob/master/MicroAlgues_Part04.ijm){< MicroAlgues_Part08.ijm>}

sourcecode : [part 8](#)

5. Cleaning

As we created new ROIS for each microalgae, they need to be deleted before treating the next one. This is simply done with this excerpt.

```
\begin{lstlisting}
NewROIcount = roiManager("count");
AllROIS=Array.getSequence(NewROIcount);
SelectedROIS = Array.slice(AllROIS,ROIcount,NewROIcount);
roiManager("Select", SelectedROIS);
roiManager("Delete");
\end{lstlisting}
```

`\textbf{sourcecode}` [\href{https://github.com/Arktthul/Measuring-the-membrane-porosity-on-microalgae-Under-progress/blob/master/MicroAlgues_Part09.ijm}](https://github.com/Arktthul/Measuring-the-membrane-porosity-on-microalgae-Under-progress/blob/master/MicroAlgues_Part09.ijm){< MicroAlgues_Part09.ijm>}

sourcecode : [part 9](#)

6. Final remarks

Don't forget to close the first "for loop" iterating on the index i between the different ROI defining each single microalgae with "}".

Full code can be found [here](#)

`\textbf{sourcecode}` [\href{https://github.com/Arktthul/Measuring-the-membrane-porosity-on-microalgae-Under-progress/blob/master/MicroAlgues_.ijm}](https://github.com/Arktthul/Measuring-the-membrane-porosity-on-microalgae-Under-progress/blob/master/MicroAlgues_.ijm){< MicroAlgues_.ijm>}

	Name	Average Intensity Algae	Intensity Slice1	Intensity Slice2	Intensity Slice3	Intensity Slice4	Inten
10	DextranChannel0002-0010-0739:Controle_10kD.tif - Series003_10	6.304	1.598	25.661	17.377	11.555	7.57
11	DextranChannel0003-0011-0228:Controle_10kD.tif - Series004_11	3.905	0.750	12.218	7.820	5.346	3.99
12	DextranChannel0003-0012-0317:Controle_10kD.tif - Series004_12	65.862	66.644	66.870	65.203	65.969	65.7
13	DextranChannel0003-0013-0504:Controle_10kD.tif - Series004_13	37.005	35.022	36.138	34.292	37.511	40.4
14	DextranChannel0003-0014-0577:Controle_10kD.tif - Series004_14	5.521	0.539	16.780	12.184	7.831	4.64
15	DextranChannel0003-0015-0613:Controle_10kD.tif - Series004_15	5.408	0.800	16.080	10.683	8.120	5.21
16	DextranChannel0003-0016-0753:Controle_10kD.tif - Series004_16	6.225	0.712	20.098	13.734	9.850	6.20
17	DextranChannel0004-0017-0059:Controle_10kD.tif - Series005_17	10.406	9.769	9.740	9.428	10.166	11.7
18	DextranChannel0004-0018-0190:Controle_10kD.tif - Series005_18	2.864	0.372	9.578	5.409	3.141	2.51
19	DextranChannel0004-0019-0195:Controle_10kD.tif - Series005_19	5.812	0.848	18.961	11.448	7.817	4.74
20	DextranChannel0004-0020-0231:Controle_10kD.tif - Series005_20	52.943	53.014	55.374	52.607	53.036	49.8
21	DextranChannel0004-0021-0303:Controle_10kD.tif - Series005_21	50.422	57.286	55.461	47.317	55.100	51.5
22	DextranChannel0004-0022-0344:Controle_10kD.tif - Series005_22	3.710	0.497	13.452	7.306	4.168	2.60
23	DextranChannel0004-0023-0394:Controle_10kD.tif - Series005_23	32.893	32.961	49.007	42.361	36.527	30.6
24	DextranChannel0004-0024-0376:Controle_10kD.tif - Series005_24	4.942	0.431	17.859	9.274	5.611	4.16
25	DextranChannel0004-0025-0397:Controle_10kD.tif - Series005_25	5.256	0.694	19.081	10.030	6.816	5.00




Figure 3: Final Table and Graph representation of the extracted measurements

Between analysis, it is important to start with a clean state of ImageJ. If closing all the images at once is easy using close all command under the file menu, closing the non image windows can take a long time. The Roi manager need to be emptied as well. To do so, I use this macro that runs independantly. It will close every logs and tables but the macro window.

```
list = getList("window.titles");
  for (i=0; i<list.length; i++){
    winame = list[i];
    selectWindow(winame);
    run("Close");
  }
```